
Subtypes for Specifications:
Predicate Subtyping in PVS

John Rushby Sam Owre N. Shankar

Abstract—A specification language used in the context of
an effective theorem prover can provide novel features that
enhance precision and expressiveness. In particular, type-
checking for the language can exploit the services of the
theorem prover. We describe a feature called “predicate
subtyping” that uses this capability and illustrate its utility
as mechanized in PVS.

I. Introduction

FOR programming languages, type systems and their
associated typecheckers are intended to ensure the ab-

sence of certain undesirable behaviors during program ex-
ecution [1]. The undesired behaviors generally include un-
trapped errors such as adding a boolean to an integer,
and may (e.g., in Java) encompass security violations. If
the language is “type safe,” then all programs that can
exhibit these undesired behaviors will be rejected during
typechecking.

Execution is not a primary concern for specification
languages—indeed, they usually admit constructs, such as
quantification over infinite domains or equality at higher
types, that are not effectively computable—but typecheck-
ing can still serve to reject specifications that are erroneous
or undesirable in some way. For example, a minimal expec-
tation for specifications is that they should be consistent:
an inconsistent specification is one from which some state-
ment and its negation can both be derived; such a speci-
fication necessarily allows any property to be derived and
thus fails to say anything useful at all. The first system-
atic type system (now known as the “Ramified Theory of
Types”) was developed by Russell [2] to avoid the incon-
sistencies in näıve set theory, and a simplified form of this
system (the “Simple Theory of Types,” due to Ramsey [3]
and Church [4]) provides the foundation for most specifica-
tion languages based on higher-order logic. If a specifica-
tion uses no axioms (beyond those of the logic itself), then
typechecking with respect to such a type system guaran-
tees consistency. However, the consistency of specifications
(such as 2 in Section III) that include extra-logical axioms
cannot be checked algorithmically in general, so the best

that a typechecker can do in the presence of axioms is to
guarantee “conservative extension” of the other parts of
the specification (i.e., roughly speaking, that it does not
introduce any new inconsistencies).

Since their presence weakens the guarantees provided
by typechecking, it is desirable to limit the use of axioms
and to prefer those parts of the specification language for
which typechecking ensures conservative extension. Un-
fortunately, those parts are usually severely limited in
expressiveness and convenience, often being restricted to
quantifier-free (though possibly recursive) definitions that
have a strongly constructive flavor; such specifications may
resemble implementations rather than statements of re-
quired properties, and proofs about them may require in-
duction rather than ordinary quantifier reasoning. Thus, a
worthwhile endeavor in the design of type systems for speci-
fication languages is to increase the expressiveness and con-
venience of those constructions for which typechecking can
guarantee conservative extension, so that the drawbacks
to a definitional style are reduced and resort to axioms is
needed less often.

In developing type systems for specification languages,
we can consider some design choices that are not avail-
able for programming languages. In particular, a speci-
fication language is meant to be part of an environment
that includes an effective theorem prover, so it is feasi-
ble to contemplate that typechecking can rely on general
theorem proving, and not be restricted to the trivially de-
cidable properties that are appropriate for programming
languages.

“Predicate subtypes” are one example of the opportuni-
ties that become available when typechecking can use the-
orem proving.1 Predicate subtypes can be used to check
statically for violations such as division by zero or out-
of-bounds array references, and can also express more so-
phisticated consistency requirements. Typechecking with
respect to predicate subtypes is done by proof obligation
(verification condition) generation.

In the following sections we will use simple examples to
explain what predicate subtypes are, and to demonstrate
their utility in a variety of situations. The examples illus-
trating the use of predicate subtypes are all drawn from
the PVS specification language [6].

II. Predicate Subtypes

Types in specification languages are often interpreted as
sets of values, and this leads to a natural association of
subtype with subset: one type is a subtype of another if

1Another is consistency checking for tabular specifications [5].

the set interpreting the first is a subset of that interpreting
the second. In this treatment (found, for example, in Mizar
[7]) the natural numbers are a subtype of the integers, but
the subtyping relation does not characterize those integers
that are natural numbers. Predicate subtypes provide such
a tightly bound characterization by associating a predicate
or property with the subtype. For example, the natural
numbers are the subtype of the integers characterized by
the predicate “greater than or equal to zero.” Predicate
subtypes can help make specifications more succinct by al-
lowing information to be moved into the types, rather than
stated repeatedly in conditional formulas. For example, in-
stead of

∀(i, j: int): i ≥ 0 ∧ j ≥ 0 ⊃ i+j ≥ i

(where we use ⊃ for logical implication) we can say

∀(i, j: nat): i+j ≥ i

because i ≥ 0 and j ≥ 0 are recorded in the type nat for
i and j.

Theorem proving can be required in typechecking some
constructions involving predicate subtypes. For example,
if half is a function that requires an even number (defined
as one equal to twice some integer) as its argument, then
the formula

∀(i: int): half(i+i+2) = i+1

is well-typed only if we can prove that the integer expres-
sion i+i+2 satisfies the predicate for the subtype even:
that is, if we can discharge the following proof obligation.

1∀(i:int):∃(j: int): i+i+2 = 2×j

Predicate subtypes seem a natural idea and often appear,
in inchoate form, in informal mathematics. Similar ideas
are also seen in formalized specification notations where,
for example, the datatype invariants of VDM [8, Chapter 5]
have much in common with predicate subtypes. However,
datatype invariants are part of VDM’s mechanisms for
specifying operations in terms of pre- and post-conditions
on a state, rather than part of the type system for its logic.
Predicate subtypes are fully supported as part of a specifi-
cation logic by the Nuprl [9], ABEL [10], Raise [11], Veritas
[12], and PVS [6] verification systems. Predicate subtypes
arose independently in these systems (in PVS, they came
from its predecessor, Ehdm, whence they were introduced
from the ANNA notation [13] by Friedrich von Henke, who
was involved in the design of both), and there are differ-
ences in their uses and mechanization. In Nuprl, for ex-
ample, all typechecking relies on theorem proving, whereas
in PVS there is a firm distinction between conventional
typechecking (which is performed algorithmically) and the
proof obligations (they are called Type Correctness Con-
ditions, or TCCs in PVS) engendered by certain uses of
predicate subtyping.

The circumstances in which proof obligations are gen-
erated, and other properties of predicate subtypes are de-
scribed in the remainder of this paper. The examples use
PVS notation, which is briefly introduced in the following
section.

PVS and its Notation for Predicate Subtypes

PVS is a higher-order logic in which the simple theory of
types is augmented by dependent types and predicate sub-
types. Type constructors include functions, tuples, records,
and abstract data types (freely generated recursive types)
such as trees and lists. A large collection of standard the-
ories is provided in libraries and in the PVS “prelude”
(which is a built-in library). The PVS system includes
an interactive theorem prover that can be customized with
user-written “strategies” (similar to tactics and tacticals
in LCF-style provers), and that provides rather powerful
automation in the form of decision procedures (e.g., for
ground equality and linear arithmetic over both integers
and reals) integrated with a rewriter [14,15].

As noted, some constructions involving predicate sub-
types generate TCCs (proof obligations); these are not de-
cidable in general as there are no constraints on the pred-
icates used to induce subtypes. However, many of the
TCCs encountered in practice do fall within a class that
is decided by the automated procedures of PVS. In other
cases, the user must develop suitable proofs interactively;
this arrangement provides the flexibility of arbitrary type
constraints without loss of automation on the decidable
ones. Proof of TCCs can be postponed, but the system
keeps track of all undischarged proof obligations and the
affected theories and theorems are marked as incomplete.

A PVS specification is a collection of theories. Each
theory takes a list of theory parameters and provides a
list of declarations or definitions for variables, individ-
ual constants, types, and formulas. Types in PVS are
built starting with uninterpreted types and primitive in-
terpreted ones, such as bool, int (integer), nat (natu-
ral number), and various other numeric types. Record
types are given as a list of label/type pairs: for example,
[# age: nat, years employed: nat #]. Tuples, such as
[nat, bool], are similar to records except that fields are
accessed by the order of their appearance rather than by
labels. Function types are introduced by specifying their
domain and range types: for example, binary arithmetic
operations such as addition and multiplication have the
type [[real, real]→real], which can also be written as
[real, real→real].

Functions (and predicates, which are simply functions
with range type bool) can be defined using λ-notation,
so that the predicate that recognizes even integers can be
written as follows (it is a PVS convention that predicates
have names ending in “?”).2

even?: [int→bool] = λ(i:int):∃(j:int): i = 2×j

However, the following “applicative” form is exactly equiv-
alent and is generally preferred.

even?(i:int): bool = ∃(j:int): i = 2×j

The strictness of the type hierarchy ensures that the princi-
ple of comprehension is sound in higher-order logic: that is,

2For ease of reading, the typeset rendition of PVS is used here; PVS
can generate this automatically using its LATEX-printer. PVS uses the
Gnu Emacs editor as a front end and its actual input is presented in
ASCII.

predicates and sets can be regarded as essentially equiva-
lent.3 PVS therefore also allows set notation for predicates,
so that the following definition is equivalent to the previous
two.

even?: [int→bool] = {i:int | ∃(j:int): i = 2×j}

Viewed as a predicate, the test that an integer x is even is
written even?(x); viewed as a set it is written x ∈ even?.
These are notational conveniences; semantically, the two
forms are equivalent.

Predicates induce a subtype over their domain type; this
subtype can be specified using set notation (overloading the
previously introduced use of set notation to specify pred-
icates), or by enclosing a predicate in parentheses. Thus,
the following are all semantically equivalent, and denote
the type of even integers.

even: TYPE = {i:int | ∃(j:int): i=2×j}
even: TYPE = (even?)
even: TYPE = (λ(i:int):∃(j:int): i=2×j)
even: TYPE = ({i:int | ∃(j:int): i=2×j})

III. Discovering Errors
with Predicate Subtypes

PVS makes no a priori assumptions about the cardinal-
ity of the sets that interpret its types: they may be empty,
finite, or countably or uncountably infinite. When an un-
interpreted constant is declared, however, we need to be
sure that its type is not empty (otherwise we have a con-
tradiction). This cannot be checked algorithmically when
the type is a predicate subtype, so an “existence TCC” is
generated that obliges the user to prove the fact.4 Thus
the constant declaration

c: even

generates the following proof obligation, which requires
nonemptiness of the even type to be demonstrated.

c_TCC1: OBLIGATION ∃(x: even): TRUE

The existence TCC is a potent detector of erroneous spec-
ifications when higher (i.e., function and predicate) types
are involved, as the following example illustrates.

Suppose we wish to specify a function that returns the
minimum of a set of natural numbers presented as its ar-
gument. Definitional specifications for this function are
likely to be rather unattractive—certainly involving a re-
cursive definition and possibly some concrete choice about
how sets are to be represented. An axiomatic specification,
on the other hand, seems very straightforward: we simply
state that the minimum is a member of the given set, and
no larger than any other member of the set. In PVS this
could be written as follows.

3All members of a set are of the same type in higher-order logic; this
notion of “set” differs from that used in set theory where {a, {a}},
for example, is a valid set.

4If the constant is interpreted (e.g., c: even = 2), then the proof
obligation is to show that its value satisfies the corresponding predi-
cate (e.g., ∃ (j: int): 2 = 2×j).

2min(s: setof[nat]): nat

simple_ax: AXIOM
∀(s:setof[nat]): min(s)∈ s ∧ ∀(n: nat): n∈ s ⊃ min(s)≤ n

Here, the first declaration gives the “signature” of the func-
tion, stating that it takes a set of natural numbers as its
argument and returns a natural number as its value. The
axiom simple ax then formalizes the informal specification
in the obvious way, and seems innocuous enough. However,
as many readers will have noticed, this axiom harbors an
inconsistency: it states that the function returns a member
of its argument s—but what if s is empty?

How could predicate subtypes alert us to this inconsis-
tency? Well, as noted earlier, sets and predicates are equiv-
alent in higher-order logic, so that a set s of natural num-
bers is also a predicate on the natural numbers, and thereby
induces the predicate subtype (s) comprising those natural
numbers that satisfy (or, viewed as a set, are members of)
s. Thus we can modify the signature of our min function
to specify that it returns, not just a natural number, but
one that is a member of the set supplied as its argument.5

min(s: setof[nat]): (s)

Now this declaration is asserting the existence of a func-
tion having the given signature and, in higher-order logic,
functions are just constants of “higher” type. Because we
have asserted the existence of a constant, we need to ensure
that its type is nonempty, so PVS generates the following
TCC.

min_TCC1: OBLIGATION ∃(x: [s: setof[nat]→ (s)]): TRUE

Inspection, or fruitless experimentation with the theorem
prover, should convince us that this TCC is unprovable
and, in fact, false.6 We are thereby led to the realization
that our original specification is unsound, and the min func-
tion must not be required to return a member of the set
supplied as its argument when that set is empty.

We have a choice at this point: we could either return
to the original signature for the min function in 2 and
weaken its axiom appropriately, or we could strengthen the
signature still further so that the function simply cannot
be applied to empty sets. The latter choice best exploits
the capabilities of predicate subtyping, so that is the one
used here. The predicate that tests a set of natural num-
bers for nonemptiness is written nonempty?[nat] in PVS,
so the type of nonempty sets of natural numbers is writ-
ten (nonempty?[nat]), and the strict signature for a min

function can be specified as follows.

min(s: (nonempty?[nat])): (s)

5This is an example of a “dependent” type: it is dependent because
the type of one element (here, the range of the function) depends on
the value of another (here, the argument supplied to the function).
Dependent typing is essential to derive the full utility of predicate
subtyping. It is discussed in more detail in Section VII.

6A function type is nonempty if its range type is nonempty or its
domain type is empty. Here the domain type is nonempty (be care-
ful not to confuse emptiness of the domain type, setof[nat], with
emptiness of the argument s), so we need to be sure that the range
type, (s), is also nonempty—which it is not, when s is empty.

This declaration generates the following TCC

min_TCC1: OBLIGATION ∃(x: [s: (nonempty?[nat])→ (s)]): TRUE

which can be discharged by instantiating x with the choice
function for nonempty types that is built-in to PVS.7

With its signature taken care of, we can now return to
the axiom that specifies the essential property of the min

function. First, notice that the first conjunct in the ax-
iom simple ax shown in 2 is unnecessary now that this
constraint is enforced in the range type of the function.
Next, notice that the implication in the second conjunct
can be eliminated by changing the quantification so that n
ranges over only members of s, rather than over all natural
numbers. This leads to the following more compact axiom.

min_ax: AXIOM ∀(s:(nonempty?[nat])), (n: (s)): min(s) ≤ n

Satisfied that this specification is correct (as indeed it
is), we might be tempted to make the “obvious” next step
and define a max function dually.

max(s: (nonempty?[nat])): (s)
max_ax: AXIOM ∀(s:(nonempty?[nat])), (n: (s)): max(s) ≥ n

This apparently small extension introduces another incon-
sistency: for what if the set s is infinite? Infinite sets of
natural numbers have a minimum element, but not a max-
imum. Let us see how predicate subtypes could help us
avoid this pitfall.

Using predicate subtyping, we can eliminate the axiom
max ax and add the property that it expresses to the range
type of the max function as follows.

3max(s: (nonempty?[nat])): {x: (s) | ∀(n: (s)): x ≥ n}

This causes PVS to generate the following TCC to ensure
nonemptiness of the function type specified for max.

max_TCC1: OBLIGATION
∃(x1: [s: (nonempty?[nat])→{x: (s)|∀(n: (s)): x≥ n}]): TRUE

Observe that by moving what was formerly specified by
an axiom into the specification of the range type, we are
using PVS’s predicate subtyping to mechanize generation
of proof-obligations for the axiom satisfaction problem.

We begin the proof of this TCC by instantiating x1 with
the (built-in) choice function choose, applied to the predi-
cate {x: (s) | ∀(n: (s)): x ≥ n} that appears as the
range type.

(INST + "λ(s:(nonempty?[nat])):
choose({x: (s) | ∀(n: (s)): x ≥ n})")

PVS proof commands are given in Lisp syntax; the first
term identifies the command (here “INST” for instantiate),
the second generally indicates those formulas in the se-
quent (see below) to which the command should be ap-
plied (+ means “any formula in the consequent part of the
sequent”), and any required PVS text is enclosed in quotes.
The next two proof commands

7We need to demonstrate the existence of a function that takes
a nonempty set of natural numbers as its argument and returns a
member of that set as its value. Choice functions, which are discussed
in Section IV, have exactly this property.

(GRIND :IF-MATCH NIL)
(REWRITE "forall_not")

then reduce the TCC to the following proof goal. (s!1
and x!1 are the Skolem constants corresponding to the
quantified variables s and x in the original formula).

4[-1] x!1 ≥ 0
[-2] s!1(x!1)

|-------
{1} ∃(x: (s!1)):∀(n: (s!1)): x ≥ n

This is a “sequent,” which is the manner in which PVS
presents the intermediate stages in a proof. In general,
there is a collection of “antecedent” formulas (here two)
above the sequent line (|-------), and a collection (here,
only one) of “consequents” below; the sequent is true if
the conjunction of formulas above the line implies the dis-
junction of formulas below (if there are no formulas be-
low the line then we need a contradiction among those
above). PVS proof commands transform the current se-
quent to one or more simpler (we hope) sequents whose
truth implies the original one. The three proof commands
shown earlier respectively instantiate an existentially quan-
tified variable (INST), perform quantifier elimination, def-
inition expansion, and invoke decision procedures (GRIND;
the annotation :IF-MATCH NIL instructs the prover not to
attempt instantiation of variables), and apply a rewrite rule
(REWRITE; the rule concerned comes from the PVS prelude
and changes a ∀ . . .¬ . . . above the line into an ∃ . . . below
the line, which makes it easier to read). Once again, inspec-
tion, or fruitless experimentation with the theorem prover,
should persuade us that the goal 4 is unprovable (we are
asked to prove that any nonempty set of natural numbers
has a largest element) and thereby reveals the flaw in our
specification.

The flaw revealed in max might cause us to examine a
specification for min given in the same form as 3 to check
that it does not have the same problem. This min spec-
ification generates a TCC that reduces to a goal similar
to 4 (with ≤ substituted for ≥ in the consequent) but,
unlike the max case, this goal is true, and can be proved
by appealing to the well-foundedness (i.e., absence of infi-
nite descending chains) of the less-than ordering on natural
numbers.

With the significance of well-foundedness now revealed
to us, we might attempt to specify a generic min function:
one that is defined over any type, with respect to a well-
founded ordering on that type.

5minspec[T: TYPE, <: (well_founded?[T])]: THEORY
BEGIN

IMPORTING equalities[T]

min((s: (nonempty?[T]))): {x: (s) |∀(i: (s)): x< i ∨ x = i}
END minspec

This specification introduces a general min function in the
context of a theory parameterized by an arbitrary (and
possibly empty) type T, and a well-founded ordering < over
that type. Notice how predicate subtyping is used in the
formal parameter list of this theory to specify that < must
be well-founded (the predicate well founded? is defined in

the PVS prelude). A proof obligation to check satisfaction
of this requirement will be generated whenever the theory
is instantiated. Observe that the specification has been
adjusted a little to separate the < and = cases that were
combined into ≤ for the special case of natural numbers.

Typechecking this specification results in the following
TCC, requiring us to demonstrate that the function type
asserted for min is nonempty.

6min_TCC1: OBLIGATION
∃(x1: [s: (nonempty?[T])→{x: (s)|∀(i: (s)): x< i ∨ x = i}]):

TRUE

As before, we begin the proof of this TCC by instantiat-
ing it with the choice function choose, applied to the predi-
cate {x: (s) | ∀(i: (s)): x<i ∨ x=i} that appears as
the range type.

(INST + "λ(s:(nonempty?[T])):
choose({x: (s) | ∀(i: (s)): x<i ∨ x=i})")

This discharges the original proof obligation, but choose

requires its argument to be nonempty, so the prover gen-
erates a new TCC subgoal to establish this fact.

7min_TCC1 (TCC):

|-------
{1} ∀(s: (nonempty?[T])):

nonempty?[(s)]({x: (s) |∀(i: (s)): x< i ∨ x = i})

This is asking us to demonstrate the existence of a minimal
element for any nonempty set s (more precisely, it is asking
us to demonstrate the nonemptiness of the set of all such
minimal elements). Now the type specified for < requires
it to be a well-founded ordering, and we can introduce
this knowledge into the proof by the command (TYPEPRED

"<"). The command (GRIND :IF-MATCH NIL) then pro-
duces the following simplified sequent.

{-1} s!1(x!1)
{-2} ∀(p: pred[T]):

(∃(y: T): p(y))
⊃ (∃(y: (p)): (∀(x: (p)): (¬ x < y)))

{-3} ∀(x: (s!1)):¬ ∀(i: (s!1)): x < i ∨ x = i
|-------

Here, the formula {-2} is expressing the well-foundedness
of <; instantiating the variable p with s!1 and giving a
few more interactive commands, we arrive at the following
sequent (this is one of two subgoals generated; the other is
trivial).

[-1] s!1(x!1)
|-------

{1} i!1 < y!1
{2} y!1 < i!1
{3} y!1 = i!1

For the specialized min function on natural numbers, the
decision procedures completed the proof at this point, but
here we recognize that this goal is not true in general, and
we need the additional assumption that the relation < be
trichotomous (i.e., one of the three consequents must hold,
as they do on the natural numbers). Once again, predicate
subtypes have led us to discover an error in our specifica-
tion. We can exit the prover, modify the specification 5

to stipulate that the theory parameter < must be of type
well ordered?[T] (a well-ordering is one that is both well-
founded and trichotomous) and rerun the proof of the TCC.
This time we are successful.

Given the generic theory, we can recover min on the nat-
ural numbers by the instantiation min[nat,<]. Because of
the subtype constraint specified for the second formal pa-
rameter to the theory, PVS generates a TCC requiring us to
establish that < on the natural numbers is a well-ordering.
This is easily done, but min[nat,>] correctly generates a
false TCC (this theory instantiation is equivalent to our
previous attempt to specify a max function on the natu-
rals). However, the TCC for min[{i: int | i< 0},>]
(i.e., the max function on the negative integers) is true and
provable.

The examples in this section illustrate how a uniform
check for nonemptiness of the type declared for a constant
leads to the discovery of several quite subtle errors in the
formulation of an apparently simple specification. In our
experience, the same benefit accrues in larger specifica-
tions.

IV. Automating Proofs with Predicate Subtypes

A theorem prover is needed to discharge the proof obli-
gations engendered by predicate subtyping. Conversely,
however, predicate subtypes provide information that can
actively assist a theorem prover. In this section we illus-
trate two ways in which predicate subtypes can help auto-
mate proofs, beginning with the use of very precise range
types for functions.

A couple of the proofs in the previous section used the
“choice function” choose. PVS actually has two choice
functions defined in its prelude. The first, epsilon, is sim-
ply Hilbert’s ε operator.

epsilons [T: NONEMPTY_TYPE]: THEORY
BEGIN

p: VAR setof[T]

epsilon(p): T

epsilon_ax: AXIOM (∃x: x ∈ p) ⊃ epsilon(p) ∈ p
END epsilons

Given a set p over a nonempty type T, epsilon(p) is some
member of p, if any such exist, otherwise it is just some
value of type T. (The VAR declaration for p simply allows us
to omit its type from the declarations where it is used; PVS
formulas are implicitly universally quantified over their free
variables.)

If p is constrained to be nonempty, then we can give the
following specification for an epsilon alt function, which
is simply epsilon specialized to this situation (note that
T need not be specified as NONEMPTY TYPE in this case).

choice [T: TYPE]: THEORY
BEGIN

p: VAR (nonempty?[T])

epsilon_alt(p): T

epsilon_alt_ax: AXIOM epsilon_alt(p) ∈ p
END choice

The new choice function epsilon alt is similar to the
built-in function choose, but if we return to the proof
of min TCC1 (recall 6 ) but use epsilon alt in place of
choose, we find that in addition to the subgoal 7 , we are
presented with the following.

8|-------
{1} ∀(s: (nonempty?[T])): ∀(i: (s)):

epsilon_alt[(s)]({x: (s)|∀(i: (s)): x<i ∨ x=i}) < i
∨ epsilon_alt[(s)]({x: (s)|∀(i: (s)): x<i ∨ x=i}) = i

This subgoal requires us to prove that the value
of epsilon alt satisfies the predicate supplied as
its argument; it can be discharged by appealing to
epsilon alt ax, but the proof takes several steps and gen-
erates a further subgoal that is similar to 7 (and proved in
the same way). How is it that the choice function choose

avoids all this work that epsilon alt seems to require?
The explanation is found in the definition of choose.

p: VAR (nonempty?[T])

choose(p): (p)

This very economical definition uses a predicate subtype to
specify the property previously stated in epsilon alt ax:
namely, that the value of choose(p) is a member of p.8 But
because the fact is stated in a subtype and is directly bound
to the range type of choose, it is immediately available to
the theorem prover—which is therefore able to discharge
the equivalent to 8 internally.

A further use of subtypes to assist automation of proofs
involves encapsulation. Below is the specification for the
cardinality of a finite set.9

S: VAR finite_set
n: VAR nat

inj_set(S): (nonempty?[nat]) =
{n | ∃(f: [(S)→below(n)]): injective?(f)}

Card(S): nat = min(inj_set(S))

Here, inj set(S) is the set of natural numbers n for which
there is an injection from S to the initial segment of natu-
ral numbers smaller than n (the predicate injective? is
defined in 16 in Section VII). Card(s) is then defined as
the least such n.

This construction has the advantage of being defini-
tional, and therefore demonstrably sound, but it is inconve-
nient to work with. Consequently, the PVS library provides
numerous lemmas that are derived from the definition but
are more suitable for automated reasoning. Unfortunately,
however, the automated prover strategies may sometimes
choose to open up the definition of Card when it would be
better to rewrite with the lemmas.

8The full definition is actually choose(p): (p) = epsilon(p); this
additionally specifies that choose(p) returns the same value as
epsilon(p), which is useful in specifications that use both epsilon
and choose.

9This technique was developed by Ricky Butler and Paul Miner
of NASA Langley, and the specification is from the new finite sets
library that was largely developed by them and is distributed with
the current version of PVS.

One way to abstract away the definition of cardinality is
to use predicate subtypes to encapsulate it in the type of
a new cardinality function.

card(S): {n: nat | n = Card(S)}

This implicitly defines card because the range type is a sin-
gleton. The lemmas can then be stated in terms of card,
which is used as the main cardinality function, and auto-
mated prover strategies can be used safely because there is
no definition of card for them to open up inappropriately.

Whereas the previous section demonstrated the utility
of predicate subtypes in detecting errors in specifications,
the examples in this section demonstrate their utility in
improving the automation of proofs. When properties are
specified axiomatically, it can be quite difficult to automate
selection and instantiation of the appropriate axioms dur-
ing a proof (unless they have special forms, such as rewrite
rules). Properties expressed as predicate subtypes on the
type of a term are, however, intimately bound to that term,
and it is therefore relatively easy for a theorem prover to
locate and instantiate the property automatically. Predi-
cate subtypes also provide a form of encapsulation, so that
specifications can be written in a style that prevents the
theorem prover from opening up certain definitions.

V. Enforcing Invariants
with Predicate Subtypes

Consider a specification for a city phone book. Given
a name, the phone book should return the set of phone
numbers associated with that name; there should also be
functions for adding, changing, and deleting phone num-
bers. Here is the beginning of a suitable specification in
PVS, giving only the basic types, and a function for adding
a phone number p to those recorded for name n in phone
book B.

names, phone_numbers: TYPE
phone_book: TYPE = [names→setof[phone_numbers]]
B: VAR phone_book
n: VAR names
p: VAR phone_numbers

add_number(B, n, p): phone_book = B WITH [(n) := B(n)∪{p}]
...

Here, the WITH construction is PVS notation for func-
tion overriding: B WITH [(n) := B(n)∪{p}] is a func-
tion that has the same values as B, except that at n it has
the value B(n)∪{p}.

Now suppose we wish to enforce a constraint that the sets
of phone numbers associated with different names should
be disjoint. We can easily do this by introducing the
unused number predicate and modifying the add number

function as follows.

9unused_number(B, p): bool = ∀(n: names):¬ p ∈ B(n)

add_number(B, n, p): phone_book =
IF unused_number(B, p) THEN B WITH [(n) := B(n)∪{p}]

ELSE B ENDIF

If we had specified other functions for updating the phone
book, they would need to be modified similarly.

But where in this modified specification does it say ex-
plicitly that different names must have disjoint sets of
phone numbers? And how can we check that our specifi-
cations of updating functions such as add number preserve
his property? Both deficiencies are easily overcome with a
predicate subtype: we simply change the type phone book

to the following.

10phone_book: TYPE =
{B: [ names → setof[phone_numbers]] |

∀(n, m: names): n 6= m ⊃ disjoint?(B(n), B(m))}

This states exactly the property we require. Furthermore,
typechecking the specification 9 now causes the following
proof obligation to be generated. Similar proof obligations
would be generated for any other functions that update the
phone book.

11add_number_TCC1: OBLIGATION
∀(B, n, p): unused_number(B, p)

⊃ ∀(r, m: names): r 6= m
⊃ disjoint?(B WITH [(n) := B(n)∪{p}](r),

B WITH [(n) := B(n)∪{p}](m))

This proof obligation, which is discharged by three com-
mands to the PVS theorem prover, requires us to prove
that a phone book B (having the disjointness property),
will satisfy the disjointness property after it has been up-
dated by the add number function.

The proof obligation in 11 arises for the same reason as
the one in 1 : a value of the parent type has been sup-
plied where one of a subtype is required, so a proof obli-
gation is generated to establish that the value satisfies the
predicate of the subtype concerned. Here, the body of the
definition given for add number in 9 has type [names →
setof[phone numbers]], which is the parent type given
for phone book in 10 , and so the proof obligation 11 is
generated to check that it satisfies the appropriate predi-
cate.

Observe how this uniform check on the satisfaction of
predicate subtypes automatically generates the proof obli-
gations necessary to ensure that the functions on a data
type (here, phone book) preserve an invariant. In the
absence of such automation, we would have to formulate
the appropriate proof obligations manually (a tedious and
error-prone process), or construct a proof obligation gen-
erator for this one special purpose. The following section
describes how the same mechanism can alleviate difficulties
caused by partial functions.

VI. Avoiding Partial Functions
with Predicate Subtypes

Functions are primitive and total in higher-order logic,
whereas in set theory they are constructed as sets of pairs
and are generally partial. There are strong advantages in
theorem proving from adopting the first approach: it allows
use of congruence closure as a decision procedure for equal-
ity over uninterpreted function symbols, which is essential
for effective automation [16]. On the other hand, there
are functions, such as division, that seem inherently par-
tial and cause difficulty to this approach. One way out of
the difficulty is introduce some artificial value for undefined

terms such as x/0, but this is clumsy and has to be done
carefully to avoid inconsistencies. Another approach intro-
duces “undefined” as a truth value [17]; more sophisticated
approaches use “free logics” in which quantifiers range only
over defined terms (e.g., Beeson’s logic of partial terms [18];
Parnas [19] and Farmer [20] have introduced logics similar
to Beeson’s10). Both approaches have the disadvantage of
using nonstandard logics, with some attendant difficulties.
These problems have led some to argue that the discipline
of types can be too onerous in a specification language, and
that untyped set theory is a better choice [22].

Predicate subtypes offer another approach. Many partial
functions become total if their domains are specified with
sufficient precision; applying a function outside its domain
then becomes a type error, rather than something that has
to be dealt with in the logic. Predicate subtypes provide
the tool necessary to specify domains with suitable preci-
sion.

For example, division is a total function if it is typed so
that its second argument must be nonzero. In PVS this
can be specified as follows.

12nonzero_real: TYPE = {x: real | x 6= 0}
/: [real, nonzero_real→real]

Now consider the well-formedness of following formula.

13test: THEOREM ∀(x, y:real): x 6= y ⊃ (x-y)/(y-x) = -1

Subtraction is closed on the reals, so x-y and y-x are both
reals. The second argument to the division function is re-
quired to have type nonzero real; real is its parent type,
so we have the proof obligation (y-x) 6= 0, which is not
true in general. However, the antecedent to the implication
in 13 will be false when x = y, rendering the theorem true
independently of the value of the improperly typed appli-
cation of division. This leads to the idea that the proof
obligation should take account of the context in which the
application occurs, and should require only that the appli-
cation is well-typed in circumstances where its value mat-
ters. In this case, a suitable, and easily proved, proof obli-
gation is the following.

test_TCC1: OBLIGATION ∀(x, y: real): x 6= y ⊃ (y-x) 6= 0

This is, in fact, the TCC generated by PVS from the for-
mula 13 . PVS imposes a left-to-right interpretation on
formulas, and generates TCCs that ensure well-formedness
under the logical context accumulated in that order. For
example, the requirements for well-formedness of an im-
plication P ⊃ Q are that P be well-formed, and that
Q be well-formed under the assumption that P is true;
if-then-else is treated as two implications, and the rules
for disjunctions P ∨ Q and conjunctions P ∧ Q are simi-
lar to that for implication, except that for disjunctions Q
must be shown well-formed under the assumption that P is

10Farmer’s logic is used in the IMPS system [21]. IMPS generates
proof obligations to ensure definedness during proofs that are sim-
ilar to PVS’s TCCs. However, because the properties required to
discharge these are not bound to the types, many similar proof obli-
gations can arise repeatedly during a single proof; IMPS mitigates
this problem using caching.

false. Thus, PVS generates the same TCC as above when
the formula in 13 is reformulated as follows.

test: THEOREM ∀(x, y:real): x = y ∨ (x-y)/(y-x) = -1

However, the accumulation of context in left-to-right order
(which is sound, but conservative) causes PVS to generate
the unprovable TCC (y-x) 6= 0 for the following, logically
equivalent, reformulation.

14test: THEOREM ∀(x, y:real): (x-y)/(y-x) = -1 ∨ x = y

Another example of a partial function is the subp “chal-
lenge” from Cheng and Jones [23]. This function on inte-
gers is given by

subp(i, j) = if i = j then 0 else subp(i, j + 1) + 1 endif

and is undefined if i < j (when i ≥ j, subp(i, j) = i− j).
As described in an earlier paper [6, Section III], this chal-

lenge is easily handled in PVS using dependent predicate
subtyping to require that the second argument is no greater
than the first. The function is then specified as follows.

subp((i:int), (j:int | j ≤ i)): RECURSIVE nat =
IF i = j THEN 0 ELSE subp(i, j+1) + 1 ENDIF

MEASURE i-j

This generates three TCCs: one to ensure that the recur-
sive call satisfies the type specified for the arguments, one
to ensure that i-j in the MEASURE satisfies the predicate for
nat, and another to establish termination using this mea-
sure. All three are discharged automatically by the PVS
decision procedures.

The PVS formulation of subp is adequate for most pur-
poses, but the following formula (from Maharaj and Bicar-
regui [24]) reveals a limitation.

15subp_lemma: FORMULA
∀ i, j: nat: subp(i,j) = i-j ∨ subp(j,i) = j-i

This formula is true in some treatments of partial functions,
but generates false TCCs and is unacceptable to PVS.

We have considered using symmetric rules for TCC gen-
eration so that examples such as 14 and 15 can be ac-
cepted: the left side of an expression would be typechecked
in the context of the right side, as well as vice-versa, and
the expression would be considered type-correct if either
proof obligation can be discharged. However, we decided
not to adopt this treatment for reasons of simplicity and
efficiency. Since most specifications are written to be read
from left to right (for the convenience of human readers),
the conservatism of the left-to-right interpretation is sel-
dom a problem in practice. (An exception is when PVS
specifications are mechanically generated from some other
representation.)

Predicate subtypes also yield an elegant treatment of re-
cursive datatypes. The stack datatype, for example, can
be specified as consisting of a constructor empty and an-
other constructor push with accessors top and pop. In
PVS, this is specified concisely as follows.

10The traditional notation for the second bound variable is
(j: {j: int | j≤ i}); PVS also allows the less redundant form
used here.

stack [T: TYPE]: DATATYPE
BEGIN

empty: empty?
push (top: T, pop:stack): nonempty?

END stack

Each constructor defines a disjoint subtype of the datatype
so that the top and pop operations are total on nonempty?

stacks and it is type-incorrect to apply them to possibly
empty? stacks. Thus, the following definition

doublepop(s: (nonempty?)): stack = pop(pop(s))

correctly generates the following unprovable and untrue
TCC.

doublepop_TCC1: OBLIGATION
∀ (s: (nonempty?[T])): nonempty?[T](pop[T](s))

In our experience, use of predicate subtypes to render
functions total is not onerous, and contributes clarity and
precision to a specification; it also provides potent error de-
tection. As another illustration of the latter, the “domain
checking” for Z specifications provided by the Z/EVES sys-
tem [25] has reportedly found errors in every Z specification
examined in this way [26]. (Domain checking is similar to
the use of predicate subtypes described in this section, but
lacks the more general benefits of predicate subtyping.)

VII. Dependent Types
and Higher-Order Subtypes

Predicate subtypes are useful for defining very refined
type dependencies through dependent typing. We have al-
ready seen a few examples of dependent typing, such as
some of the treatments of min in Section III and subp in
the previous section. Here, we illustrate its use to constrain
the fields of records to “reasonable” values, and to ensure
a natural treatment of equality.

Dependent typing can be used to constrain the type of
one field in a record according to the value of another field,
as in the following example.

employee_record: TYPE =
[# age: nat, years_employed: upto(age) #]

This record declaration constrains the years employed

field to take values that are bounded above by the value of
the age field (upto(n) is the type {i: nat | i≤ n}). This
type would thus rule out a record such as the following.

Jones: employee_record = (# age := 30, years_employed := 40 #)

The utility of the combination of dependent typing and
predicate subtyping can be further illustrated by an exam-
ple due to Carl Witty: the “implementation” of stacks as a
record consisting of a size field and an array of elements
of some type T. A simple formalization of this is the fol-
lowing record type.

stack_imp: TYPE = [# size: nat, elements: [nat→T] #]

The problem with this implementation is that two stacks
that have the same size, say n, and agree on the first n val-
ues in the elements array, can still be unequal by disagree-
ing on irrelevant array values (i.e., those beyond size).11

11Note that equality on functions is extensional: two functions f
and g of type [D→R] are equal iff for all x in D, f(x) = g(x).

This makes it awkward to formulate correctly even such
simple theorems as pop(push(x, stack)) = stack. The
problem can be solved using dependent typing to reformu-
late the specification as follows.

stack_imp: TYPE = [# size: nat, elements: [below(size)→T] #]

With this refined typing, two stacks are equal when they
have the same size and agree on the stack elements. Finite
sequences are defined similarly in the PVS prelude.

PVS is a higher-order logic, meaning that functions can
be applied to functions, and quantification can range over
function types. Consequently, predicates can be defined
over functions, and induce corresponding subtypes. For
example, the following theory from the PVS prelude defines
the predicates injective?, surjective? and bijective?

over functions from D to R.12

16functions [D, R: TYPE]: THEORY
BEGIN
f: VAR [D→R]
x, x1, x2: VAR D
y: VAR R
injective?(f): bool = ∀ x1, x2: f(x1)=(x2) ⊃ x1=x2
surjective?(f): bool = ∀ y: ∃ x: f(x) = y
bijective?(f): bool = injective?(f) ∧ surjective?(f)

These induce corresponding subtypes, allowing declara-
tions such as the following.

int2nat: (bijective?[int, nat]) =
λ (i:int): IF i>0 THEN 2*i-1 ELSE -2*i ENDIF

By the standard mechanisms, this generates a TCC requir-
ing a demonstration that the function int2nat is indeed a
bijection between the integers and the naturals.

The combination of predicate subtyping, dependent typ-
ing, and higher-order types and subtypes is a powerful one.
Higher-order subtypes can be used to introduce types such
as order-preserving or order-inverting maps, and monotone
predicate transformers. A drawback to some uses of pred-
icate subtypes, however, is that large numbers of TCCs
may be generated. In the next section we describe how
this drawback can be overcome.

VIII. Judgements

The examples given in the preceding three sections illus-
trate the proof obligations that are generated when a term
of a given type is provided where a subtype is expected.
This can lead to a proliferation of many similar proof obli-
gations. One way the PVS typechecker controls this pro-
liferation is to check whether a new TCC is subsumed by
earlier ones in the same theory: the TCC is suppressed if
it is so subsumed. Although this does remove some dupli-
cates, it is still possible to generate TCCs that differ only
in some irrelevant contextual formulas.

An effective way to minimize—and often eliminate—the
generation of trivially different TCCs for a given expres-
sion is for the user to state the needed proof obligation
once and for all, and in its strongest form. In PVS, this
is accomplished using judgements. The simplest form of

12Since a tuple type may be supplied for D, this theory is fully
general and can be instantiated for functions of any arity.

judgement states that a given constant has a specific type.
For example, we could give the judgement declaration

17JUDGEMENT 2 HAS_TYPE even

This generates an immediate TCC to show that 2 is indeed
even, but the typechecker can then make use of this fact
to avoid generating similar TCCs in any context where the
judgement is visible.

Judgements can also assert subtype constraints on the
value returned by a function in terms of those on its argu-
ments. Suppose we have the following formula declaration.

18h: FORMULA ∀ (e: even): half(e+2) = e/2 + 1

Recall that half requires an even argument. We would like
the typechecker to recognize that the result of adding two
even numbers is again even; this can be accomplished with
the following judgement declaration.13

19JUDGEMENT + HAS_TYPE [even, even → even]

Such a judgement over a function type is interpreted as a
closure condition equivalent to the following formula.

∀(e1, e2: even): even?(e1 + e2)

This is, in fact, the TCC generated by this judgement dec-
laration.14 The combination of the judgements 17 and 19

allows the PVS typechecker to determine immediately that
the application of half in 18 is well-typed.

The final kind of judgement informs the typechecker
of subtype relations that are not explicit in their con-
structions. For example, the types nonzero real and
nonzero rational are defined as follows in the PVS pre-
lude.

nonzero_real: NONEMPTY_TYPE = {r: real | r 6= 0}
nonzero_rational: NONEMPTY_TYPE = {r: rational | r 6= 0}

From this the typechecker can deduce the subtype relation-
ship between nonzero rational and rational, and hence
also the type real (since rational is defined as a sub-
type of real), but it cannot deduce a subtyping relation
between nonzero rational and nonzero real. With di-
vision typed as in 12 in Section VI, the following formula
generates a TCC to show that q is nonzero.

div_lt_1: FORMULA ∃ (q: nonzero_rational): 1/q = 2

Such proof obligations can be avoided if the desired subtype
relation is stated explicitly using following judgement.

JUDGEMENT nonzero_rational SUBTYPE_OF nonzero_real

As with other judgement declarations, this immediately
generates the necessary TCC and enlarges the collection
of facts known to the typechecker, thereby reducing the
number of TCCs generated subsequently.

13This, along with a large number of similar judgements, is in the
PVS prelude.

14A future version of PVS will allow such closure constraints to be
stated more directly as: JUDGEMENT +(e1, e2: even) HAS TYPE even

IX. Conversions

Conversions are functions that the typechecker can insert
automatically whenever there is a type mismatch. Their
purpose is to provide increased syntactic convenience in
situations involving subtyping of higher types, but we in-
troduce them by means of a simpler example.

c: [int→bool]
CONVERSION c
two: FORMULA 2

Here, since formulas must be of type boolean, the type-
checker automatically invokes the conversion and changes
the formula to c(2). This is done internally, and is only
visible to the user on explicit command and in the proof
checker. (To avoid confusion, the typechecker warns the
user if there is ever more than one applicable conversion.)

This simple kind of conversion has nothing to do with
subtypes, but standard conversions restrict and extend

do play an important role in handling subtyping on func-
tion types in PVS. The rule for subtyping of function types
is straightforward for the range type, so that [D→R1] is a
subtype of [D→R2] iff R1 is a subtype of R2. However, the
treatment of domains is less obvious. The covariant ap-
proach regards [D1→R] as a subtype of [D2→R] iff D1 is
a subtype of D2; conversely, the contravariant approach re-
gards [D1→R] as a subtype of [D2→R] iff D2 is a subtype
of D1. PVS makes a more restricted choice: the domains of
two functions must be equal for them to have a subtyping
relation (which is then determined by their range types).
This choice keeps the semantics simple (see Section X), but
prohibits some natural constructions. Consider the follow-
ing, for example.

g: [int→int]
F: [[nat→int]→bool]
F_app: FORMULA F(g)

As this stands, F app is not type-correct, because a func-
tion of type [int→int] is supplied where one of type
[nat→int] is required, and PVS requires equality on do-
main types before a subtyping relation can be considered.

However, it is clear that g naturally induces a function
from nat to int by simply restricting its domain. Such a
domain restriction is achieved by the restrict conversion
that is defined in the PVS prelude as follows.

restrict [T: TYPE, S: TYPE FROM T, R: TYPE]: THEORY
BEGIN

f: VAR [T→R]
s: VAR S
restrict(f)(s): R = f(s)
CONVERSION restrict

END restrict

The construction S: TYPE FROM T specifies that the ac-
tual parameter supplied for S must be a subtype of the one
supplied for T. The specification states that restrict(f)

is a function from S to R whose values agree with f

(which is defined on the larger domain T). Using this ap-
proach, a type correct version of F app can be written as
F(restrict[int,nat,int](g)). This is, of course, incon-
venient to read and write, so restrict is specified as a

conversion, which allows the PVS typechecker to insert it
automatically when needed, thereby providing much of the
convenience of contravariant function subtyping in this cir-
cumstance.

It is not so obvious how to expand the domain of a func-
tion in the general case, so this approach does not work so
automatically in the other direction. It does, however, work
well for the important special case of sets (or, equivalently,
predicates): a set on some type S can be extended natu-
rally to one on a supertype T by assuming that the mem-
bers of the type-extended set are just those of the original
set. Thus, if extend(s) is the type-extended version of the
original set s, we have extend(s)(x) = s(x) if x is in the
subtype S, and extend(s)(x) = false otherwise. We can
say that false is the “default” value for the type-extended
function. Building on this idea, we arrive at the following
specification for a general type-extension function.

extend [T: TYPE, S: TYPE FROM T, R: TYPE, d: R]: THEORY
BEGIN

f: VAR [S→R]
t: VAR T
extend(f)(t): R = IF S_pred(t) THEN f(t) ELSE d ENDIF

END extend

The function extend(f) has type [T→R] and is con-
structed from the function f of type [S→R] (where S is
a subtype of T) by supplying the default value d whenever
its argument is not in S (S pred is the recognizer predicate
for S). Because of the need to supply the default d, this con-
struction cannot be applied automatically as a conversion.
However, as noted above, false is a natural default for
functions with range type bool (i.e., sets and predicates),
and the following theory establishes the corresponding con-
version.

extend_bool [T: TYPE, S: TYPE FROM T]: THEORY
BEGIN

CONVERSION extend[T, S, bool, false]
END extend_bool

In the presence of this conversion, the type-incorrect for-
mula B app in the following specification

b: [nat→bool]
B: [[int→bool]→bool]
B_app: FORMULA B(b)

is automatically modified by the typechecker to become
B(extend[int,nat,bool,false](b)).

These examples illustrate the utility of conversions in
bringing some of the convenience of contravariant function
subtyping to the more restricted type system of PVS. Con-
versions are also useful (for example, in semantic encodings
of temporal logics) in “lifting” operations to apply point-
wise to sequences over their argument types.

X. Comparison with Subtypes
in Programming Languages

We know of no programming language that provides
predicate subtypes, although the annotations provided for
“extended static checking” (proving the absence of runtime
errors such as array bound violations) [27, 28] have some
similarities. Bringing the benefits of predicate subtyping

to programming languages seems a worthwhile research en-
deavor that might generalize the benefits of extended static
checking, while also providing information that could be
useful to an optimizing compiler.

Subtypes of a different, “structural,” kind are sometimes
used in type systems for programming languages to account
for issues arising in object-oriented programs [1]. In par-
ticular, a record type A that contains fields in addition to
those of a record type B is regarded as a subtype of B. The
intuition behind this kind of subtyping is rather different
than the “subtypes as subsets” intuition. Here, the idea
is that a subtype is an elaboration of a type, so that any-
where a value of a certain type is required, it should be ac-
ceptable to supply a value of a subtype of that type (e.g.,
a function that requires “points” should find a “colored
point” acceptable). Structural subtypes are characterized
by having a canonical coercion from the subtype to the su-
pertype (e.g., by dropping the extra fields from a record)
so that a supertype operation can be applied by means of
this coercion. Using this approach, some operations can be
structural subtype polymorphic—meaning that they ap-
ply uniformly to all structural subtypes of a given type.
When these ideas are applied to functions, they lead to
the “normal” or covariant subtyping on range types, but
contravariant subtyping on domain types.

We know of no specification language that provides
structural subtyping, still less combines it with predicate
subtyping. The difficulty is that reasoning about equality is
problematical in the presence of structural record subtypes
and contravariant or covariant function subtyping.

In PVS the type of the equality relation in an equa-
tion x = y can be determined simply by considering the
types of x and y (it is equality on their least common su-
pertype). For example, in the expression x/y × y = x,
where x and y are natural numbers, it is equality on ra-
tionals (because the division operator coerces the left hand
expression to be rational). This means that if x = y and
y = z are type-correct, then so is x = z, and it is true if
the other two are. Now consider contravariant subtyping
on function domains. This allows abs = id[nat], where
abs: [int→nat] is the absolute value function on the in-
tegers and id[T] is the identity function on type T (here,
the naturals). This follows by promoting abs to its con-
travariant supertype [nat→nat] and taking equality on
that type. It also allows id[nat] = id[int] by the same
reasoning, and transitivity of equality might lead us to con-
clude abs = id[int]. But there is no reason to invoke
contravariant subtyping in this final equation, since both
functions have the same domain (we do need to use co-
variant subtyping on the range), so the equality is on the
type [int→int], and the equation is false. (The equa-
tion is true if equality is interpreted on [nat→nat] but, as
noted, there is no reason to assign this type on the basis of
the arguments appearing in the expression.)

Because of this difficulty in contravariant subtyping on
function domains,15 and in order to allow equals to be freely

15Covariant subtyping on domains presents difficulties, too: it
seems to require partial functions.

substituted, we have chosen to allow function subtyping
in PVS only when the domains are equal. PVS does ex-
tend subtyping covariantly over the range types of func-
tions (e.g., [nat → nat] is a subtype of [nat → int])
and over the positive parameters to abstract data types
(e.g., list of nat is a subtype of list of int), since these
cases do not present problems. We consider predicate sub-
typing to be a basic element of a specification language—
whereas structural subtyping and subtyping on function
domains are largely syntactic conveniences that we pre-
fer to handle by mechanisms such as conversions (see also
[29]), rather than incorporate them into the type system.
Nonetheless, combining some structural subtyping (e.g., for
records) with predicate subtyping is an interesting topic for
research.

XI. Conclusion

We have illustrated a few circumstances where predicate
subtypes contribute to the clarity and precision of a specifi-
cation, to the identification of errors, and to the automation
provided in analysis of specifications and in theorem prov-
ing. There are many other circumstances where predicate
subtypes provide benefit, and they have been used to ex-
cellent effect by several users of PVS (see, for example, the
PVS bibliography [30] and the links from its Web page).
We hope the examples we have presented do convey the
value of predicate subtyping in specification languages, and
suggest their possible utility in programming languages.

